Kissoide (Zissoide)

Univ.-Prof. Dr.-Ing. habil. Josef BETTEN
RWTH University Aachen
Mathematical Models in Materials Science and Continuum Mechanics
Augustinerbach 4-20
D-52056 A a c h e n , Germany

betten@mmw.rwth-aachen.de

```
> Kissoide_des_Diokles[Efeublatt]:=x^3-(a-x)*y^2=0;
```

$$Kissoide\_des\_Diokles_{Efeublatt} := x^3 - (a - x)\, y^2 = 0$$

Damit kann ein Efeublatt dargestellt werden. Als Beispiel sei a = 1 gewählt:

```
> restart:
> with(plots,implicitplot):
> alias(co=color,th=thickness):
> plot1:=implicitplot(x^3+(x-1)*y^2=0,
  x=0..1/2,y=0..1/sqrt(2),numpoints=5000,co=black,th=3):
> plot2:=implicitplot(x^3+(x-1)*y^2=0,
  x=0..1/2,y=-1/sqrt(2)..0,numpoints=1000,co=black,th=3):
> plot3:=implicitplot(x^2+y^2=1/2,
  x=1/2..sqrt(2),y=-1/sqrt(2)..1/sqrt(2),co=black,th=3):
> plots[display](plot1,plot2,plot3);
```



\>

Im Bild ist eine Kissoide dargestelt (plot1 für y>0 und plot2 für y<0). Die beiden Zweige sind mit einem Kreisbogen (plot3) verbunden. Dadurch entsteht ein Efeublatt. Der Name Kissoide kommt aus dem Griechischen: kissos = Efeu. Die Form (Gleichung) einer Kissoide wird durch zwei vorgegebene Kurven k[1] und k[2] (Leitkurven) bestimmt, die man auch zur Konstruktion der entsprechenden Kissoide mit Zirkel und Lineal zu Grunde legt.

Im obigen Beispiel sind k[1] ein Kreis mit dem Radius a/2 und k[2] eine senkrechte Tangente:

```
> k[1]:=(x-a/2)^2 + y^2 = a^2/4;    k[2]:= x = a;
```

$$k_1 := \left(x - \frac{a}{2}\right)^2 + y^2 = \frac{a^2}{4}$$

$$k_2 := x = a$$

```
> restart:
> with(plots,implicitplot):
> alias(co=color,H=Heaviside,th=thickness):
> # a = 1,
> p[1]:=implicitplot(x^3+(x-1)*y^2=0,
  x=0..1,y=0..3/2,numpoints=5000,co=black,th=3):
> p[2]:=implicitplot({(x-1/2)^2+y^2=1/4,y-4*x/3=0},
  x=0..1,y=0..3/2,co=black):
> p[3]:=plot((3/2)*H(x-1),x=0..1.001,y=0..3/2,
  co=black,scaling=constrained):
> p[4]:=plot([[1,4/3],[9/25,12/25]],
  style=point,symbol=circle,symbolsize=30):
> p[5]:=plot([[16/25,64/75]],style=point,
  symbol=cross,symbolsize=60):
> p[6]:=plots[textplot]({[0.3,1.3,`A -> B` = `0 -> P`],
  [8/25,13/25,`A`],[0.95,4/3,`B`],[19/25,64/75,`P(x,y)`]}):
> plots[display](seq(p[k],k=1..6));
```




```
>
```

Aus der Definition AB = OP erhält man unter Berücksichtigung der Leitkurven k[1] und k[2] die Gleichung der Kissoide

```
> x^3-(a-x)*y^2=0;
```

$$x^3 - (a - x)\, y^2 = 0$$

die man in Polarkoordinaten (r, phi) gemäß

```
> r(phi):=a*(sin(phi))^2/cos(phi);
```

$$\mathrm{r}(\phi) := \frac{a \sin(\phi)^2}{\cos(\phi)}$$

darstellen kann. Eine rationale Parameterdarstellung ist folgendermaßen möglich:

```
> x(t):=a*t^2/(1+t^2);    y(t):=a*t^3/(1+t^2);
```

$$\mathrm{x}(t) := \frac{a\, t^2}{1 + t^2}$$

$$\mathrm{y}(t) := \frac{a\, t^3}{1 + t^2}$$

mit t = y(t) / x(t) = tan(phi).

Diokles, ein griechischer Mathematiker, erfand um 100 v. Chr. zur Lösung der "Würfelvolumenverdoppelung" die nach ihm benannte Kissoide.

Nach einer Legende befragten die Bewohner der Insel Delos während einer Pestepidemie (um 400 v. Chr.) das Orakel von Delphi um Rat. Sie wurden aufgefordert, das Volumen des würfelformigen Altars im Tempel des Apollon zu verdoppeln.
Die Seitenlänge "a" müsste auf b = 2^(1/3)*a vergrößert werden, was nur mit Zirkel und Lineal nicht möglich ist, da die irrationale Zahl 2^(1/3) nicht durch ganze Zahlen ausgedrückt werden kann.

Legt man als Leitkurven eine Ellipse und Asymptote

```
> restart:
> k[1]:=(x-a/2)^2+(y/b)^2=a^2/4;    k[2]:=x=a;
```

$$k_1 := \left(x - \frac{a}{2}\right)^2 + \frac{y^2}{b^2} = \frac{a^2}{4}$$

$$k_2 := x = a$$

zugrunde, so erhält man mit den Parametern a = 1 und b = 1/2 folgendes Bild:

```
> with(plots,implicitplot):
> alias(H=Heaviside,th=thickness,co=color):
> # Beispiel:  a = 1 und b = 1/2
> p[1]:=implicitplot(x^3+(x-1)*(y/(1/2))^2=0,x=0..1,y=0..1,
  numpoints=5000,th=3,co=black):
> p[2]:=implicitplot({(x-1/2)^2+(y/(1/2))^2=1/4,x=1,y=0.8*x,y=1},
  x=0..1,y=0..1,scaling=constrained,co=black):
> p[3]:=plot([[1,0.8],[0.281,0.225]],
  style=point,symbol=circle,symbolsize=30):
> p[4]:=plot([[0.72,0.575]],
  style=point,symbol=cross,symbolsize=60):
> p[5]:=plots[textplot]({[0.3,0.8,`A -> B` = `O -> P`],
  [0.25,0.25,`A`],[0.95,0.8,`B`],[0.8,0.575,`P(x,y)`]}):
> plots[display](seq(p[k],k=1..5));
```




```
>
```

Im nächsten Beispiel wird als Leitkurven der Einheitskreis und eine waagerechte Linie zu Grund gelegt. Damit erhält man folgende Kissoide:

```
> restart:
> Leitkurve_k[1]:=x^2+y^2=1;    Leitkurve_k[2]:=y-b=0;
```

$$Leitkurve\_k_1 := x^2 + y^2 = 1$$

$$Leitkurve\_k_2 := y - b = 0$$

```
> Kissoide:=(x^2+y^2)*(y+b)^2-a^2*y^2=0;
```

$$Kissoide := (x^2 + y^2)\,(y + b)^2 - a^2\,y^2 = 0$$

Als Beispiel sei a = 1 und b =1/3 bzw. b = -1/3 gewählt:

```
> eq(x,y):=subs({a=1,b=1/3},Kissoide);
```

$$\mathrm{eq}(x, y) := (x^2 + y^2)\left(y + \frac{1}{3}\right)^2 - y^2 = 0$$

```
> Eq(x,y):=subs({a=1,b=-1/3},Kissoide);
```

$$\mathrm{Eq}(x, y) := (x^2 + y^2)\left(y - \frac{1}{3}\right)^2 - y^2 = 0$$

```
> with(plots,implicitplot):
> alias(th=thickness,co=color):
> p[1]:=implicitplot(eq(x,y),x=-1..1,y=0..1,
  numpoints=5000,scaling=constrained,th=3,co=black):
> p[2]:=implicitplot(Eq(x,y),x=-1..1,y=-1..0,
  numpoints=5000,th=3,co=black):
> p[3]:=implicitplot({x^2+y^2=1,y=1/3,y=-1/3,y=2.5*x},
  x=-1..1,y=-1..1,co=black):
> p[4]:=plot([[0.133,1/3],[0.37,0.93]],
  style=point,symbol=circle,symbolsize=30):
> p[5]:=plot([[0.2381,0.5951]],
  style=point,symbol=cross,symbolsize=60):
```

```
> p[6]:=plots[textplot]({[-0.5,0.75,`AB`=`OP`],
  [0.15,0.28,`A`],[0.42,0.95,`B`],[0.3,0.6,`P(x,y)`],
  [0.9,0.5,`k1`],[0.75,0.38,`k2`]},align=RIGHT):
> plots[display](seq(p[k],k=1..6));
```




```
>
```

Koordinaten des Punktes P(x,y):

```
> x[P]:=solve(subs(y=2.5*x,eq(x,y)),x);   y[P]:=2.5*%;
```

$$x_P := 0., 0., 0.2380573430, -0.5047240097$$

$$y_P := 0., 0., 0.5951433575, -1.261810024$$

Darin sind nur die Werte größer NULL relevant:

```
> (x[p], y[p])=(0.2381, 0.5951);
```

$$(x_p, y_p) = (0.2381, 0.5951)$$

```
>
```

Legt man die Leitlinien

```
> restart:
> Leitlinie_k[1]:=x=a;    Leitlinie_k[2]:=y=b;
```

$$Leitlinie\_k_1 := x = a$$

$$Leitlinie\_k_2 := y = b$$

zu Grunde, so erhält man folgende Kissoide:

```
> Kissoide:=b*x/(a-x)-y=0;
```

$$Kissoide := \frac{b\,x}{a-x} - y = 0$$

Es sei a = 1 und b = 1/3 bzw. b = -1/3 gewählt:

```
> eq(x,y):=x/3/(1-x)-y=0;
```

$$\mathrm{eq}(x, y) := \frac{x}{3\,(1-x)} - y = 0$$

```
> Eq(x,y):=-x/3/(1-x)-y=0;
```

$$\mathrm{Eq}(x,y) := -\frac{x}{3\,(1-x)} - y = 0$$

```
> with(plots,implicitplot):
> alias(H=Heaviside,th=thickness,co=color):
> p[1]:=implicitplot(eq(x,y),x=0..6/7,y=0..2,th=3,co=black):
> p[2]:=implicitplot(Eq(x,y),x=0..6/7,y=-2..0,th=3,co=black):
> p[3]:=plot({-1/3,1/3,-2,2,-2*H(x-1),2*H(x-1),1.8*x},
  x=0..1.001,y=-2..2,co=black):
> p[4]:=plot([[1/5.4,1/3],[1,1.8]],
  style=point,symbol=circle,symbolsize=30):
> p[5]:=plot([[4.4/5.4,4.4/3]],
  style=point,symbol=cross,symbolsize=60):
> p[6]:=plots[textplot]({[0.3,1.8,`AB`=`OP`],[0.16,0.45,`A`],
  [0.95,1.81,`B`],[0.85,4.4/3,`P(x,y)`],[0.95,1,`k1`],
  [0.9,0.45,`k2`]},align=RIGHT):
> plots[display](seq(p[k],k=1..6));
```




```
>
```

Im folgenden Beispiel werden als Leikurven der Einheitskreis und die Kreise eq(x,y) und Eq(x,y) gewählt:

```
> restart:
> eq(x,y):=(x-1/2)^2+y^2-1/4=0;
```

$$\mathrm{eq}(x,y) := \left(x-\frac{1}{2}\right)^2 + y^2 - \frac{1}{4} = 0$$

```
> Eq(x,y):=(x+1/2)^2+y^2-1/4=0;
```

$$\mathrm{Eq}(x,y) := \left(x+\frac{1}{2}\right)^2 + y^2 - \frac{1}{4} = 0$$

```
> Kissoide[x>0]:=x^2+y^2-sqrt(x^2+y^2)+x=0;
```

$$Kissoide_{0<x} := x^2 + y^2 - \sqrt{x^2+y^2} + x = 0$$

```
> Kissoide[x<0]:=x^2+y^2-sqrt(x^2+y^2)-x=0;
```

$$Kissoide_{x<0} := x^2 + y^2 - \sqrt{x^2 + y^2} - x = 0$$

```
> with(plots,implicitplot):
> alias(th=thickness,co=color,bl=black):
> p[1]:=implicitplot(eq(x,y),x=0..1,y=-1..1,co=bl):
> p[2]:=implicitplot(Eq(x,y),x=-1..0,y=-1..1,co=bl):
> p[3]:=implicitplot(x^2+y^2-1=0,x=-1..1,y=-1..1,
  scaling=constrained,co=bl):
> p[4]:=implicitplot(Kissoide[x>0],x=0..1,y=-1..1,
  numpoints=50000,th=3,co=bl):
> p[5]:=implicitplot(Kissoide[x<0],x=-1..0,y=-1..1,
  numpoints=50000,th=3,co=bl):
> plots[display](seq(p[k],k=1..5));
```



\>

Im nächsten Beispiel dienen der Einheiskreis Eq(x,y) und eine Ellipse eq(x,y) zur Erzeugung einer
Kissoide K(x,y):

```
> x^2+y^2=a^2;   (x/a)^2+(y/b)^2=1;
```

$$x^2 + y^2 = a^2$$

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$$

Daraus gewinnt man mit der Definition AB := OP die entsprechende Kissoide:

```
> a^2*b^2*(x^2+y^2)=(a^2*y^2+b^2*x^2)*(sqrt(x^2+y^2)-a)^2;
```

$$a^2 b^2 (x^2 + y^2) = (a^2 y^2 + b^2 x^2) (\sqrt{x^2 + y^2} - a)^2$$

\>

Beispielsweise sei a = 1 , und b = 1/2 gewählt:

```
> restart:
> Eq(x,y):=x^2+y^2-1=0;    eq(x,y):=x^2+4*y^2-1=0;
```

$$\mathrm{Eq}(x, y) := x^2 + y^2 - 1 = 0$$

$$\mathrm{eq}(x, y) := x^2 + 4 y^2 - 1 = 0$$

```
> K(x,y):=x^2+y^2-(4*y^2+x^2)*(sqrt(x^2+y^2)-1)^2=0;
```

$$K(x,y) := x^2+y^2-(4\,y^2+x^2)\,(\sqrt{x^2+y^2}-1)^2=0$$

```
> with(plots,implicitplot):
> alias(th=thickness,co=color):
> p[1]:=implicitplot(K(x,y),x=-1..1,y=-1..1,
  numpoints=40000,th=3,co=black):
> p[2]:=implicitplot({Eq(x,y),eq(x,y)},x=-1..1,y=-1..1,
  numpoints=1000,scaling=constrained,co=black):
> plots[display](p[1],p[2]);
```




```
>
```

Mit den Parametern a = 1 , b = 1/3 erhält man folgendes Bild:

```
> Eq(x,y):=x^2+y^2-1=0;   eq(x,y):=x^2+9*y^2-1=0;
```

$$\mathrm{Eq}(x,y) := x^2+y^2-1=0$$

$$\mathrm{eq}(x,y) := x^2+9\,y^2-1=0$$

```
> K(x,y):=(1/9)*(x^2+y^2)-((1/9)*x^2+y^2)*(sqrt(x^2+y^2)-1)^2=0;
```

$$K(x,y) := \frac{x^2}{9}+\frac{y^2}{9}-\left(\frac{x^2}{9}+y^2\right)(\sqrt{x^2+y^2}-1)^2=0$$

```
> with(plots,implicitplot):
> alias(th=thickness,co=color):
> p[1]:=implicitplot(K(x,y),x=-1..1,y=-1..1,
  numpoints=40000,th=3,co=black):
> p[2]:=implicitplot({Eq(x,y),eq(x,y),y=3*x/2},x=-1..1,y=-1..1,
  numpoints=1000,scaling=constrained,co=black):
> p[3]:=plot([[0.22,0.33],[0.55,0.83]],
  style=point,symbol=circle,symbolsize=30):
> p[4]:=plot([[1/3,1/2]],
  style=point,symbol=cross,symbolsize=60):
> p[5]:=plots[textplot]({[-0.4,3/4,`AB`=`OP`],
```

```
[0.2,0.4,`A`],[0.65,0.83,`B`],[0.5,1/2,`P(x,y)`]}):
> plots[display](seq(p[k],k=1..5));
```



>

Die obigen Beispiele zeigen, wie hilfreich die Software MAPLE ist. Auf derartige "mathematische Formelmanipulations-Programme" kann man heute nicht mehr verzichten.

>